

Using and Making Modules

Max Levatich & Stephen A. Edwards

Columbia University

Fall 2025



Using Modules

Import every name from a module:

```
import Data.List  -- Includes, e.g., nub
```

```
numUniques :: Eq a => [a] -> Int
```

```
numUniques = length . nub
```

In GHCi,

```
ghci> :m + Data.List           -- Import one module
```

```
ghci> :m + Data.List Data.Map  -- Import multiple modules
```

```
ghci> import Data.Char
```

Import Variants

```
import Data.List (nub, sort)           -- Only nub and sort  
import Data.List hiding (nub, sort)  -- All but nub and sort  
import qualified Data.List             -- Data.List.nub, etc.  
import qualified Data.List as L       -- L.nub, L.sort, etc.
```

```
ghci> :m + Data.List
```

```
ghci> intersperse '*' "MASH"  
"M*A*S*H"
```

```
ghci> intercalate ", " ["Foo","Bar","Baz"]  
"Foo, Bar, Baz"
```

```
ghci> transpose [[1,2,3],[4,5,6],[7,8,9]]  
[[1,4,7],[2,5,8],[3,6,9]]
```

```
ghci> concat ["PFP ", "is ", "fun"]  
"PFP is fun"
```

```
ghci> concatMap (replicate 3) [1..3]  
[1,1,1,2,2,2,3,3,3]
```

```
ghci> and [True, False, True]
```

```
False
```

```
-- and = foldr (&&) True
```

```
ghci> and [True, True]
```

```
True
```

```
ghci> or [True, False, True]
```

```
True
```

```
-- or = foldl (||) False
```

```
ghci> any (==4) [1..5]
```

```
True
```

```
-- any p = or . map p
```

```
ghci> all (>4) [5..10]
```

```
True
```

```
-- all p = and . map p
```

```
ghci> all (<=4) [5..10]
```

```
False
```

```
ghci> take 5 $ iterate (*2) 1  
[1,2,4,8,16]
```

```
ghci> splitAt 3 "pfprocks"  
("pfp","rocks")
```

```
ghci> takeWhile (<10) [1..]  
[1,2,3,4,5,6,7,8,9]
```

-- Prefix of list

```
ghci> dropWhile (<5) [1..10]  
[5,6,7,8,9,10]
```

-- Suffix of list

```
ghci> span (<5) [1..10]  
([1,2,3,4],[5,6,7,8,9,10])
```

-- Prefix/suffix split

```
ghci> sort [8,5,3,2,1,6,4,2]
[1,2,2,3,4,5,6,8]
```

```
ghci> group [1,1,1,2,2,1,1,1,1,5,5,4,3,3]
[[1,1,1],[2,2],[1,1,1,1],[5,5],[4],[3,3]]
```

```
ghci> maxRun = maximum . map length . group
ghci> maxRun [1,1,1,2,2,1,1,1,1,5,5,4,3,3]
4
```

```
ghci> inits "whoa!"
["","w","wh","who","whoa","whoa!"]
```

```
ghci> tails "whoa!"
["whoa!","hoa!","oa!","a!","!",""]
```

```
ghci> let s = "whoa" in zip (inits s) (tails s)
[("", "whoa"), ("w", "hoa"), ("wh", "oa"), ("who", "a"), ("whoa", "")]
```

Searching Lists

```
isPrefixOf :: Eq a => [a] -> [a] -> Bool
isPrefixOf [] _ = True
isPrefixOf _ [] = False
isPrefixOf (x:xs) (y:ys) = x == y && isPrefixOf xs ys
```

```
ghci> "PFP" `isPrefixOf` "PFP Rocks!"
True
ghci> "PFP" `isPrefixOf` "PHP Rocks!"
False
ghci> :set prompt "> "
> search needle haystack = any (isPrefixOf needle) (tails haystack)
> search "fun" "PFP is fun, dontcha know"
True
> search "fun" "Columbia"
False
```

Data.List calls it `isInfixOf` instead of `search`. There is also `isSuffixOf`

Partition and Quicksort Revisited

```
ghci> msg = "He Is Daring, Dumb, and Educated, Nancy"
ghci> partition (`elem` ['A'..'Z']) msg
("HIDDEN","e s aring, umb, and ducated, ancy")
```

```
import Data.List ( partition )
```

```
quicksort :: Ord a => [a] -> [a]
```

```
quicksort [] = []
```

```
quicksort (p:xs) = quicksort prefix ++ [p] ++ quicksort suffix
                  where (prefix,suffix) = partition (<p) xs
```

```
ghci> :l quicksort3
```

```
[1 of 1] Compiling Main
```

```
( quicksort3.hs, interpreted )
```

```
Ok, one module loaded.
```

```
ghci> quicksort "the quick brown fox jumps over the lazy dog"
"      abcdeefghhijklmnooopqrrsttuuvwxyz"
```

Lists as Text

```
ghci> lines "first\nsecond\nthird\nfourth"  
["first","second","third","fourth"]
```

```
ghci> unlines ["one","two","three"]  
"one\ntwo\nthree\n"
```

```
ghci> words "The Quick Brown Fox Jumps"  
["The","Quick","Brown","Fox","Jumps"]
```

```
ghci> unwords ["My","gosh","it's","full","of","stars"]  
"My gosh it's full of stars"
```

Lists as Sets: Assumes Unique But Unordered

```
ghci> nub [1,3,2,4,3,2,1,2,3,4,3,2,1]
[1,3,2,4] -- Duplicates removed, unordered

ghci> nub "the quick brown fox jumps over the lazy dog"
"the quickbrownfxjmpsvlazydg"

ghci> delete 'e' "Stephen"
"Stphen" -- Delete the first matching element

ghci> ([1..10] ++ [1..3]) \\ [2,5,9]
[1,3,4,6,7,8,10,1,2,3] -- List difference: delete first matching

ghci> "the quick brown fox" `union` ['a'..'z']
"the quick brown foxadgjlmpsvyz"

ghci> "the quick brown fox" `intersect` ['a'..'m']
"heickbf"

ghci> insert 'p' "almost"
"almopst" -- To last position where it's <=; maintains sorted order
```

```
genericLength    :: Num i => [a] -> i
genericTake      :: Integral i => i -> [a] -> [a]
genericDrop      :: Integral i => i -> [a] -> [a]
genericSplitAt   :: Integral i => i -> [a] -> ([a], [a])
genericIndex     :: Integral i => [a] -> i -> a
genericReplicate :: Integral i => i -> a -> [a]
```

```
nubBy            :: (a -> a -> Bool) -> [a] -> [a]
deleteBy         :: (a -> a -> Bool) -> a -> [a] -> [a]
deleteFirstsBy   :: (a -> a -> Bool) -> [a] -> [a] -> [a]
unionBy          :: (a -> a -> Bool) -> [a] -> [a] -> [a]
intersectBy      :: (a -> a -> Bool) -> [a] -> [a] -> [a]
groupBy          :: (a -> a -> Bool) -> [a] -> [[a]]
```

```
sortBy          :: (a -> a -> Ordering) -> [a] -> [a]
insertBy        :: (a -> a -> Ordering) -> a -> [a] -> [a]
maximumBy       :: Foldable t => (a -> a -> Ordering) -> t a -> a
minimumBy       :: Foldable t => (a -> a -> Ordering) -> t a -> a
```

Data.Char: Character Type Predicates

```
isAscii,      isLatin1,      isControl,  
isAsciiUpper, isAsciiLower,  
isPrint,      isSpace,       isUpper,  
isLower,      isAlpha,       isDigit,  
isOctDigit,   isHexDigit,    isAlphaNum,  
isPunctuation, isSymbol      :: Char -> Bool
```

```
ghci> import Data.Char  
ghci> all isHexDigit "18deadBEEF"  
True  
ghci> all isHexDigit "gosh"  
False  
ghci> map generalCategory " \t\nA9?|"  
[Space,Control,Control,UppercaseLetter,DecimalNumber,  
 OtherPunctuation,MathSymbol]
```

Data.Char: Conversion Functions

```
ghci> map toUpper "the quick brown fox"
"THE QUICK BROWN FOX"

ghci> map toLower "THE QUICK Brown FoX"
"the quick brown fox"

ghci> map digitToInt "09afBC"
[0,9,10,15,11,12]    -- Hex digits allowed

ghci> map intToDigit [4,2,10,15]
"42af"              -- Inverse of digitToInt

ghci> map ord " !ABab"
[32,33,65,66,97,98] -- ASCII/Unicode values

ghci> map chr [71,101,116,32,66,101,110,116]
"Get Bent"          -- Inverse of ord
```

Association Lists: Slow, Straightforward

```
phoneBook =  
  [("Jenny", "867-5309")      , ("Morris", "777-9311")  
   , ("Alessia", "273-8255")  , ("Tina", "606-0842")  
   , ("Alicia", "489-4608")   , ("Glenn", "736-5000") ]  
lookup :: Eq k => k -> [(k, v)] -> Maybe v -- in Prelude  
lookup _ [] = Nothing  
lookup kk ((k,v):l) | kk == k    = Just v  
                   | otherwise = lookup kk l
```

```
ghci> lookup "Alicia" phoneBook  
Just "489-4608"      -- Alicia is one of the keys  
ghci> lookup "Jenny" phoneBook  
Just "867-5309"  
ghci> lookup "Marty" phoneBook  
Nothing
```

```
ghci> import qualified Data.Map as Map

ghci> :t Map.fromList
Map.fromList :: Ord k => [(k, a)] -> Map.Map k a -- Ordered keys

ghci> Map.fromList [("Jenny","837-5306"),("Alicia","489-4608")]
fromList [("Alicia","489-4608"),("Jenny","837-5306")]

ghci> Map.empty
fromList [] -- The empty map

ghci> Map.insert "Alicia" "489-4608" Map.empty
fromList [("Alicia","489-4608")] -- Add a pair

ghci> fromList' = foldr (\(k,v) m -> Map.insert k v m) Map.empty
ghci> Map.null Map.empty
True -- Is the map empty?

ghci> Map.null $ Map.fromList [(1,1)]
False

ghci> Map.size $ Map.fromList [(1,1),(2,3)]
2 -- Number of pairs
```

```
ghci> Map.singleton "Jenny" "867-5309"  
fromList [("Jenny","867-5309")]  
ghci> Map.insert 1 "one" $ Map.singleton 0 "zero"  
fromList [(0,"zero"),(1,"one")]
```

```
ghci> phoneMap = Map.fromList phoneBook  
ghci> Map.lookup "Jenny" phoneMap  
Just "867-5309"  
ghci> Map.lookup "Freddy" phoneMap  
Nothing  
ghci> Map.member "Alicia" phoneMap  
True
```

```
ghci> Map.map (*10) $ Map.fromList [(2,1),(3,5),(1,8)]  
fromList [(1,80),(2,10),(3,50)]      -- Applied to values
```

```
ghci> Map.filter odd $ Map.fromList [(x,x+3) | x <- [0..8]]  
fromList [(0,3),(2,5),(4,7),(6,9),(8,11)] -- Filter values
```

```
ghci> phoneMap = Map.fromList phoneBook
ghci> Map.keys phoneMap
["Alessia","Alicia","Jenny","Morris","Tina"]

ghci> Map.elems phoneMap
["273-8255","489-4608","867-5309","777-9311","606-0842"]

ghci> Map.toList phoneMap
[("Alessia","273-8255"),("Alicia","489-4608"), -- Sorted
 ("Jenny","867-5309"),("Morris","777-9311"),
 ("Tina","606-0842")]

ghci> :set +m
ghci> let dups = [(1,1),(1,20),(2,5),(1,300),(3,8),(3,80)]
ghci| in Map.fromListWith (+) dups
fromList [(1,321),(2,5),(3,88)] -- Duplicate key's values added
```

```
ghci> import qualified Data.Set as Set
ghci> :t Set.fromList
Set.fromList :: Ord a => [a] -> Set.Set a

ghci> set1 = Set.fromList "the quick brown fox jumps over"
ghci> set2 = Set.fromList "pack my box with five dozen"
ghci> set1
fromList " bcefghijklmnopqrstuvw"    -- Unique, sorted
ghci> set2
fromList " abcdefghikmnoptvwxyz"      -- Unique, sorted
ghci> Set.union set1 set2
fromList " abcdefghijklmnopqrstvwxyz" -- in set1 or set2
ghci> Set.intersection set1 set2
fromList " bcefghikmnoptvw"           -- in set1 and set2
ghci> Set.difference set1 set2
fromList "jqrsu"                      -- in set1 but not set2
ghci> Set.difference set2 set1
fromList "adyz"                       -- in set2 but not set1
```

```
ghci> Set.null Set.empty
True
ghci> Set.null $ Set.fromList [3,4,5,5,4,3]
False
ghci> Set.size $ Set.fromList [3,4,5,5,4,3]
3
ghci> Set.singleton 42
fromList [42]
ghci> Set.insert 2 $ Set.insert 4 $ Set.singleton 1
fromList [1,2,4]
ghci> Set.delete 7 $ Set.fromList [1..10]
fromList [1,2,3,4,5,6,8,9,10]
ghci> 5 `Set.member` Set.fromList [1..10]
True
ghci> 0 `Set.member` Set.fromList [1..10]
False
```

```
ghci> :set prompt "> "
```

```
> Set.fromList [2..4] `Set.isSubsetOf`      Set.fromList [0..10]  
True
```

```
> Set.fromList [2..4] `Set.isSubsetOf`      Set.fromList [2..4]  
True
```

```
> Set.fromList [2..4] `Set.isProperSubsetOf` Set.fromList [2..4]  
False
```

```
> Set.fromList [2..4] `Set.isSubsetOf`      Set.fromList [0..3]  
False
```

```
> Set.map (2^) $ Set.fromList [1..5]  
fromList [2,4,8,16,32]
```

```
> Set.filter odd $ Set.fromList [0..10]  
fromList [1,3,5,7,9]
```

Writing a Module: Geometry.hs

```
module Geometry
( sphereVolume      -- Exported names
, cubeVolume
) where

sphereVolume :: Float -> Float
sphereVolume radius = (4.0 / 3.0) * pi * (radius ^ 3)

cubeVolume :: Float -> Float
cubeVolume side = cuboidVolume side side side

cuboidVolume :: Float -> Float -> Float -> Float
cuboidVolume a b c = rectangleArea a b * c

rectangleArea :: Float -> Float -> Float      -- Internal only
rectangleArea a b = a * b
```

Using the Geometry Package

```
ghci> :l Geometry
[1 of 1] Compiling Geometry          ( Geometry.hs, interpreted )
Ok, one module loaded.
*Geometry> :show modules
Geometry          ( Geometry.hs, interpreted )
*Geometry> :reload
Ok, one module loaded.

*Geometry> sphereVolume 10.0
4188.7905
*Geometry> cubeVolume 2
8.0
```

Breaking up Modules

Create

Geom/Sphere.hs

Geom/Cube.hs

Geom/Cuboid.hs

```
ghci> :l Geom.Sphere Geom.Cube
[1 of 3] Compiling Geom.Cuboid      ( Geom/Cuboid.hs, interpreted )
[2 of 3] Compiling Geom.Cube        ( Geom/Cube.hs, interpreted )
[3 of 3] Compiling Geom.Sphere      ( Geom/Sphere.hs, interpreted )
Ok, three modules loaded.
*Geom.Sphere> Geom.Cube.volume 2.0
8.0
```

Geom/Sphere.hs

```
module Geom.Sphere
( volume
, area
) where

volume :: Float -> Float
volume radius = (4.0 / 3.0) * pi * (radius ^ 3)

area :: Float -> Float
area radius = 4 * pi * (radius ^ 2)
```

Geom/Cuboid.hs

```
module Geom.Cuboid
( volume
, area
) where

volume :: Float -> Float -> Float -> Float
volume a b c = rectangleArea a b * c

area :: Float -> Float -> Float -> Float
area a b c = rectangleArea a b * 2 +
              rectangleArea a c * 2 +
              rectangleArea c b * 2

rectangleArea :: Float -> Float -> Float
rectangleArea a b = a * b
```

Geom/Cube.hs

```
module Geom.Cube
( volume
, area
) where

import qualified Geom.Cuboid as Cuboid

volume :: Float -> Float
volume side = Cuboid.volume side side side

area :: Float -> Float
area side = Cuboid.area side side side
```